
Quickstart
    

This guide will help you get started with XFsm. You will learn how to create a state machine, create an actor from the state machine, send events to the actor, and use it outside of an actor.
Info
The quickest way to start with XFsm is to check out the Basic Payment Machine. There you will find an example of a machine that covers the basics of XFsm.

  
    
  
  Installing XFsm


XFsm is a declarative state management and orchestration library for Elixir.
def deps do
  [
    {:xfsm, "~> 0.3"}
  ]
end

  
    
  
  Create a Machine


In XFsm, a machine is a module that contains all the logic for the actor. In this example, we will create a simple toggle machine that can be in one of two states: on or off. The press event will toggle the state between on and off.
defmodule LightbulbSwitch do
  use XFsm.Actor
  use XFsm.Machine

  initial(:off)

  state :off do
    on :press do
      target(:on)
    end
  end

  state :on do
    on :press do
      target(:off)
    end
  end
end
Just by looking at the declaration, you can get a sense of how it works. To familiarize yourself with the foundational concepts, read our introduction to state machines guide.

  
    
  
  Start an Actor and Send Events


An actor is a running process that can receive events, send events to other actors, and change its behavior based on the events it receives.
alias XFsm.{Actor, Snapshot}

# Here, we start the actor.
{:ok, pid} = LightbulbSwitch.start_link()

# Grab a snapshot of the actor's state.
%Snapshot{state: :off} = Actor.snapshot(pid)

# Send the `press` event to the actor.
:ok = Actor.send(pid, %{type: :press})

# Notice that the state of the actor has now changed?!
%Snapshot{state: :on} = Actor.snapshot(pid)

# Send another `press` event to the actor.
:ok = Actor.send(pid, %{type: :press})

# Now we're back to the initial state.
%Snapshot{state: :off} = Actor.snapshot(pid)
Info
Ensure your machine utilizes the use XFsm.Actor behavior. Additionally, XFsm actors are GenServers under the hood but have a more restricted scope of behavior.

  
    
  
  Context Data


Context is how you store data within a state machine actor.
defmodule ToggleMachine do
  use XFsm.Actor
  use XFsm.Machine

  import XFsm.Actions

  initial(:inactive)
  context(%{count: 0})

  state :inactive do
    on :toggle do
      target(:active)
    end
  end

  state :active do
    # Increment `count` every time we transition into this state.
    entry(assigns(%{count: &(&1.context.count + 1)}))

    on :toggle do
      target(:inactive)
    end
  end
end

  
    
  
  Machine Input


Machine Input is how initial data can be passed to a machine actor. Meanwhile, Guards are used to conditionally allow or disallow transitions.
defmodule ToggleMachine do
  use XFsm.Actor
  use XFsm.Machine

  import XFsm.Actions

  initial(:inactive)
  context(%{input: input}, do: %{count: 0, max_count: input.max_count})

  state :inactive do
    on :toggle do
      target(:active)
      guard(%{context: context}, do: context.count < context.max_count)
    end
  end

  state :active do
    entry(assigns(%{count: &(&1.context.count + 1)}))

    on :toggle do
      target(:inactive)
    end
  end
end

alias XFsm.Actor

{:ok, pid} = ToggleMachine.start_link(input: %{max_count: 10})

Actor.subscribe(pid, &IO.inspect/1)

Actor.send(pid, %{type: :toggle})


  

    
Install XFsm
    

XFsm has zero dependencies and runs in any Beam environment. You can install XFsm by visiting the Hex, you will find the latest version and how to install using supported package managers.


  

    
Cheatsheet
    

Use this cheatsheet to quickly lookup the APIs for XFsm.

  
    
  
  Examples



  
    
  
  Creating a state machine


defmodule ToggleMachine do
  use XFsm.Actor
  use XFsm.Machine

  initial(:active)
  context(%{count: 0})

  state :active do
    entry(assigns(%{count: &(&1.context.count +1)}))

    on :toggle, do: target(:inactive)
  end

  state :inactive do
    on :toggle, do: target(:active)
  end
end

alias XFsm.Actor

{:ok, pid} = ToggleMachine.start_link()

Actor.subscribe(pid, &IO.puts/1)

Actor.send(pid, %{type: :toggle})
# logs 'inactive' with context %{count: 1}

Actor.send(pid, %{type: :toggle})
# logs 'active' with context %{count: 2}

Actor.send(pid, %{type: :toggle})
# logs 'inactive' with context %{count: 2}

  
    
  
  Guards


defmodule ToggleMachine do
  use XFsm.Actor
  use XFsm.Machine

  initial(:active)
  context(%{activate?: false})

  state :inactive do
    on :toggle do
      target(:active)
      guard(:toggle?)
    end

    on :toggle do
      action(:notify_not_allowed)
    end
  end

  state :active do
    on :toggle do
      target(:inactive)
      guard(%{
        method: :after_time?,
        params: %{time: "00:00"}
      })
    end
  end

  defg toggle?(%{context: context}), do: context.activate?

  defg after_time?(_, %{time: time}) do
    now = DateTime.utc_now()
    [hour, minute] = String.split(time, ":")
    {hour, ""} = Integer.parse(hour)
    {minute, ""} = Integer.parse(minute)

    now.hour > hour and now.minute > minute
  end

  defa notify_not_allowed(%{context: context}) do
    IO.puts("Cannot be toggled")
    context
  end
end

  
    
  
  Actions


defmodule ToggleMachine do
  use XFsm.Actor
  use XFsm.Machine

  initial(:active)

  state :active do
    entry(:activate)
    exit(:deactivate)

    on :toggle do
      target(:inactive)
      action(:notify)
    end
  end

  state :inactive do
    on :toggle do
      target(:active)
      action(%{
        method: :notify,
        params: %{message: "Some notification"}
      })
    end
  end

  defa activate(_) do
    # ...
  end

  defa deactivate(_) do
    # ...
  end

  defa notify(_,  %{message: _}) do
    # ...
  end
end


  

    
What are State Machines?
    

State machines are models used to visualize how an object changes its state based on external events.
This model is essential in software engineering because it helps capture the various states of an entity, the events that trigger changes, and the transitions between these states. Additionally, it assists in identifying impossible states and undesirable transitions.
Some benefits of using a state machine include:
	They are simple to understand and can be easily communicated across different levels of an organization.
	They are easy to maintain and test because the states and transitions are deterministic.
	They provide a concise way to model complex application workflows.


  
    
  
  States


A state describes the condition of a machine at any given time.
For example, a lightbulb is either on or off; it can never be both simultaneously. Furthermore, it is impossible for it to be neither on nor off.
---
title: Lightbulb
---
flowchart LR

on ~~~~ off
In the example above, there are only two states, which are finite.
When a machine starts, it first enters the initial state. A machine cannot have multiple initial states because it wouldn't be able to determine which state to enter.
Info
A state machine can only be in one state at a time.
Read more about states

  
    
  
  States in XFsm


In XFsm, states are defined using the state attribute.
defmodule Lightbulb do
  use XFsm.Machine

  initial(:off)

  state :off do
  end

  state :on do
  end
end

  
    
  
  Transitions


The process of how a lightbulb goes from off to on is referred to as transitions. Events trigger transitions; when a known event occurs, the machine transitions to the next state.
The lightbulb alternates between on and off through the turn_on and turn_off events.
Transitions are deterministic; each combination of state and event always points to the same next state.
---
title: Lightbulb
---
stateDiagram-v2

[*] --> off
off --> on: turn_on
on --> off: turn_off
With its two finite states and transitions, this lightbulb process is a Finite State Machine. A state machine is used to describe the behavior of something. The machine outlines the thing’s states and the transitions between them. It’s considered a Finite State Machine because it has a finite number of states.
Read more about transitions and events

  
    
  
  Events and Transitions in XFsm


Events and transitions are defined in XFsm state machines inside the on property of a state.
defmodule Lightbulb do
  use XFsm.Machine

  initial(:off)

  state :off do
    on :turn_on, do: target(:on)
  end

  state :on do
    off :turn_off, do: target(:off)
  end
end

  
    
  
  Final State


Most processes with states will have a final state, which is the last state when the process is completed.

  
    
  
  Self-transition


Self-transition happens when an event occurs, but the transition returns to the same state. The transition arrows exits and re-enter the same state.
A helpful way to describe a self-transition is “doing something, not going somewhere” in the process.
stateDiagram-v2

state "state" as a

[*] --> a
a --> a: transition
Read more about self transitions


  

    
XFsm
    

[image: Dialyzer] [image: Test] [image: Codecov]
XFsm is a declarative finite state machine library for Elixir.
It uses event-driven programming, state machines and actors to handle complex logic in predictable and robust ways.
It provides very easy to use APIs which makes looking at a declaration very easy to understand.

  
    
  
  Installation


The package can be installed by adding xfsm to your list of dependencies in mix.exs:
def deps do
  [
    {:xfsm, "~> 0.5.0"}
  ]
end

  
    
  
  Documentation


API documentation is available at https://hexdocs.pm/xfsm

  
    
  
  Create a simple machine


defmodule Counter do
  use XFsm.Actor
  use XFsm.Machine

  import XFsm.Actions

  context(%{count: 0})

  on :inc do
    action(assigns(%{count: & &1.context.count + 1}))
  end

  on :dec do
    action(assigns(%{count: & &1.context.count - 1}))
  end

  on :set do
    action(assigns(%{count: & &1.event.value}))
  end
end

alias XFsm.Actor

{:ok, pid} = Counter.start_link()

Actor.subscribe(pid, fn %{context: %{count: count}} ->
  IO.puts(count)
end)

Actor.send(pid, %{type: :inc})
# logs 1

Actor.send(pid, %{type: :dec})
# logs 0

Actor.send(pid, %{type: :set, value: 10})
# logs 10
---
title: Counter machine
---
flowchart LR

inc --> machine
dec --> machine
set --> machine
machine --- inc
machine --- dec
machine --- set

machine@{shape: diam, label: "Machine"}
dec@{shape: event, label: "DEC\nassigns(...)"}
inc@{shape: event, label: "INC\nassigns(...)"}
set@{shape: event, label: "SET\nassigns(...)"}

  
    
  
  Create a complex machine


defmodule TicTacToe do
  use XFsm.Actor
  use XFsm.Machine

  import XFsm.Actions

  defmodule Board do
    defstruct squares: {nil, nil, nil, nil, nil, nil, nil, nil, nil}

    @tags [:x, :o]

    def square(%__MODULE__{} = board, index)
        when is_integer(index) and index >= 1 and index <= 9 do
      elem(board.squares, index - 1)
    end

    def empty?(%__MODULE__{} = board, index), do: square(board, index) == nil

    def has_empty_square?(%__MODULE__{} = board) do
      Enum.any?(0..8, &(elem(board.squares, &1) == nil))
    end

    def draw?(%__MODULE__{} = board) do
      not has_empty_square?(board) and
        not (won?(board, :x) or won?(board, :o))
    end

    def won?(%__MODULE__{} = board, tag) when tag in @tags do
      %{squares: squares} = board

      siblings?(:x, squares, tag) or
        siblings?(:y, squares, tag) or
        siblings?(:z, squares, tag)
    end

    def put(%__MODULE__{} = board, index, tag)
        when is_integer(index) and index >= 1 and index <= 9 and tag in @tags do
      case empty?(board, index) do
        false ->
          board

        true ->
          squares = put_elem(board.squares, index - 1, tag)

          %__MODULE__{board | squares: squares}
      end
    end

    def to_iodata(%__MODULE__{} = board) do
      Enum.reduce(0..2, [], fn i, acc ->
        row = Enum.map_join(0..2, "|", &(elem(board.squares, i * 3 + &1) || "-"))

        Enum.concat(acc, ["|", row, "|\n"])
      end)
    end

    defp siblings?(:x, squares, tag) do
      Enum.reduce_while(0..2, false, fn i, acc ->
        fun? = &(elem(squares, i * 3 + &1) == tag)

        case Enum.all?(0..2, fun?) do
          true -> {:halt, true}
          false -> {:cont, acc}
        end
      end)
    end

    defp siblings?(:y, squares, tag) do
      Enum.reduce_while(0..2, false, fn i, acc ->
        fun? = &(elem(squares, i + &1 * 3) == tag)

        case Enum.all?(0..2, fun?) do
          true -> {:halt, true}
          false -> {:cont, acc}
        end
      end)
    end

    defp siblings?(:z, squares, tag) do
      fun? = &(elem(squares, &1) == tag)

      Enum.all?(0..8//4, fun?) or Enum.all?(2..6//2, fun?)
    end
  end

  initial(:x)
  context(%{input: i}, do: context_from_input(i))

  state :x do
    on :move do
      target(:o)
      guard(%{method: :can_move?, params: %{player: :x}})
      action(:make_move)
    end
  end

  state :o do
    on :move do
      target(:x)
      guard(%{method: :can_move?, params: %{player: :o}})
      action(:make_move)
    end
  end

  state :end do
    # You most likely want to persist the game state and terminate the
    # process, assuming you're building an online multiplayer game.
  end

  root do
    always do
      target(:end)
      guard(%{method: :won?, params: %{player: :o}})
      action(assigns(%{winner: :o}))
    end

    always do
      target(:end)
      guard(%{method: :won?, params: %{player: :x}})
      action(assigns(%{winner: :x}))
    end

    always do
      target(:end)
      guard(%{method: :drawn?, params: %{player: :x}})
    end
  end

  defg can_move?(
         %{context: %{x: x}, event: %{ref: x, square: s}} = arg,
         %{player: :x}
       )
       when is_integer(s) and s >= 1 and s <= 9 do
    %{context: %{board: board}, event: %{square: square}} = arg

    Board.empty?(board, square)
  end

  defg can_move?(
         %{context: %{o: o}, event: %{ref: o, square: s}} = arg,
         %{player: :o}
       )
       when is_integer(s) and s >= 1 and s <= 9 do
    %{context: %{board: board}, event: %{square: square}} = arg

    Board.empty?(board, square)
  end

  defg won?(
         %{self: %{state: _}, context: %{board: _}} = arg,
         %{player: player} = params
       )
       when player in [:x, :o] do
    %{self: %{state: state}, context: %{board: board}} = arg
    %{player: player} = params

    state != :end and Board.won?(board, player)
  end

  defg drawn?(%{self: %{state: _}, context: %{board: _}} = arg) do
    %{self: %{state: state}, context: %{board: board}} = arg

    state != :end and Board.draw?(board)
  end

  defa make_move(%{context: %{x: x} = c, event: %{ref: x, square: s}})
       when is_integer(s) and s >= 1 and s <= 9 do
    %{board: board} = c
    board = Board.put(board, s, :x)

    %{c | board: board}
  end

  defa make_move(%{context: %{o: o} = c, event: %{ref: o, square: s}})
       when is_integer(s) and s >= 1 and s <= 9 do
    %{board: board} = c
    board = Board.put(board, s, :o)

    %{c | board: board}
  end

  defp context_from_input(%{x: x, o: o})
       when is_reference(x) and is_reference(o) do
    %{x: x, o: o, winner: :none, board: %Board{}}
  end
end

alias XFsm.Actor

# We're using references so that a player does not make a move for the other.
# These can be user id assuming you're building an online multiplayer game.
x = make_ref()
o = make_ref()

{:ok, pid} = TicTacToe.start_link(input: %{x: x, o: o})

Actor.subscribe(pid, fn %{context: %{board: b, winner: w}} ->
  iodata = TicTacToe.Board.to_iodata(b)

  IO.puts([iodata | ["\n", "winner: ", Atom.to_string(w)]])
end)

:ok = Actor.send(pid, %{type: :move, ref: x, square: 5})
:ok = Actor.send(pid, %{type: :move, ref: o, square: 3})
:ok = Actor.send(pid, %{type: :move, ref: x, square: 6})
:ok = Actor.send(pid, %{type: :move, ref: o, square: 1})
:ok = Actor.send(pid, %{type: :move, ref: x, square: 4})

XFsm.Actor.snapshot(pid)
stateDiagram-v2

state x_if_end <<choice>>
state o_if_end <<choice>>

state "player x" as x
state "player o" as o

[*] --> x
x --> x_if_end
x_if_end --> end: if pos = draw or win
x --> o: move

o --> o_if_end
o_if_end --> end: if pos = draw or win
o --> x: move

  
    
  
  License


MIT

          @type self() :: %{pid: nil | pid(), state: nil | atom()}

